

Byte 2: Type byte = 0H
 Byte 3: Length byte = m, # of data bytes following
 Bytes 4-(3+m): Data bytes containing source program
 Byte 4+m: Checksum byte = sum of data bytes, mod 256

Additional data blocks follow only if the length byte contains 255.

The following assembly language routine indicates how SAVED programs can be loaded into memory, and may be used as a template for applications in which the user wishes to manipulate SAVED programs without using XYBASIC. The desired file name should be stored at FILNAM. The file is read from the RDR device and loaded starting at address DEST.

```

STBYTE EQU    3AH    ;start byte
TYBYTE EQU    0      ;type byte

HEADER: DB     STBYTE, TYBYTE
FILNAM: DS     8      ;ASCII filename, padded by spaces
        DB     'XYB'  ;file type

LOAD:   LXI     H,HEADER;header location to HL
        MVI     C,13   ;header length to C
LOAD1:  CALL    RDRIN   ;read a character
        CMP     M      ;compare to header character
        JNZ     LOAD    ;no match, try again
        DCR     C      ;decrement count
        JNZ     LOAD1   ;read next header character
        LXI     H,DEST  ;destination to HL
LOAD2:  CALL    RDRIN   ;read character
        CPI     STBYTE  ;compare to start byte
        JNZ     ERROR
        CALL    RDRIN   ;read another
        CPI     TYBYTE  ;compare to type byte
        JNZ     ERROR
        CALL    RDRIN   ;read length byte
        ORA     A       ;check if length 0
        RZ      ;done if length 0
        MOV     E,A     ;length to E
        INR     A
        PUSH    PSW     ;save length+1
        MVI     D,0     ;checksum to D
LOAD3:  CALL    RDRIN   ;read a source character
        MOV     M,A     ;store it
        INX     H       ;point to next destination
        ADD     D
        MOV     D,A     ;update checksum
        DCR     E       ;decrement count
        JNZ     LOAD3   ;load more characters from data block
        CALL    RDRIN   ;read the checksum
        CMP     D       ;compare to computed checksum
        JNZ     ERROR   ;checksum error
        POP     PSW     ;recover length+1
        JZ      LOAD2   ;length was 255, load more data blocks
        RET            ;otherwise LOAD is completed

ERROR:  ...           ;LOAD error routine
  
```